```
var C H I Ih K Y mc N Rd R Q pk w wr Pi Pir f b f1 f2 x1 x2 L dp A mu Phi RL Ry QEH
RyEH TP Lam M logY logRd logI logC logPi logTP logQ logRy;

varexo eA em ep eR eb;

parameters beta kappa TPs Ls epsi epsw thetap thetaw delta alpha h by taupi tauy rhoR
sR psin psik rhoA sA rhom sm rhob1 sb rhop sp Rds QEHs RyEHs Rys RLs Qs zeta Ms mcs Rs
Ks ws Ys Is Cs Lams B fs bs Ns PHIs eta rhob2;

load param_cfp17;
set_param_value('alpha',alpha);
set_param_value('beta',beta);
set_param_value('delta',delta);
set_param_value('epsi',epsi);
set_param_value('epsw',epsw);
set_param_value('kappa',kappa);
set_param_value('TPs',TPs);
set_param_value('Ls',Ls);
set_param_value('thetap',thetap);
set_param_value('thetaw',thetaw);
set_param_value('h',h);
set_param_value('by',by);
set_param_value('taupi',taupi);
set_param_value('tauy',tauy);
set_param_value('rhoR',rhoR);
set_param_value('sR',sR);
set_param_value('psin',psin);
set_param_value('psik',psik);
set_param_value('rhoA',rhoA);
set_param_value('sA',sA);
set_param_value('rhom',rhom);
set_param_value('sm',sm);
set_param_value('rhob1',rhob1);
set_param_value('rhob2',rhob2);
set_param_value('sb',sb);
set_param_value('rhop',rhop);
set_param_value('sp',sp);
set_param_value('Rds',Rds);
set_param_value('QEHs',QEHs);
set_param_value('RyEHs',RyEHs);
set_param_value('Rys',Rys);
set_param_value('RLs',RLs);
set_param_value('Qs',Qs);
set_param_value('zeta',zeta);
set_param_value('Ms',Ms);
set_param_value('mcs',mcs);
set_param_value('Rs',Rs);
set_param_value('Ks',Ks);
set_param_value('ws',ws);
set_param_value('Ys',Ys);
set_param_value('Is',Is);
set_param_value('Cs',Cs);
set_param_value('Lams',Lams);
set_param_value('B',B);
set_param_value('fs',fs);
set_param_value('bs',bs);
set_param_value('Ns',Ns);
set_param_value('PHIs',PHIs);
set_param_value('eta',eta);

model;

%%%%%%%%%%%%%%%
%%%% household non-wage
%%%%%%%%%%%%%%%
% (1) MU of consumption
Lam = (C - h*C(-1))^(-1) - beta*h*(C(+1) - h*C)^(-1);

% (2) Euler equation bonds
```

```
Lam = beta*Lam(+1)*Rd*Pi(+1)^(-1);

% (3) Euler equation capital
pk*Lam*M = beta*Lam(+1)*(R(+1) + (1-delta)*pk(+1)*M(+1));

% (4) Euler equation long bonds
Lam*Q*M = beta*Lam(+1)*Pi(+1)^(-1)*(1+kappa*Q(+1)*M(+1));

% (5) Capital accumulation
K = Ih + (1-delta)*K(-1);

% (6) Loan-in-advance
pk*Ih = Q*(f - kappa*Pi^(-1)*f(-1));

%%%%%%%%%%%%%%
%%%% household wage
%%%%%%%%%%%%%%
% (7) Reset wage
wr = (epsw/(epsw-1))*f1/f2;

% (8) f1
f1 = (w/wr)^(epsw*(1+eta))*B*H^(1+eta) +
thetaw*beta*Pi(+1)^(epsw*(1+eta))*(wr(+1)/wr)^(epsw*(1+eta))*f1(+1);

% (9) f2
f2 = Lam*H*(w/wr)^(epsw) + thetaw*beta*Pi(+1)^(epsw-1)*(wr(+1)/wr)^(epsw)*f2(+1);

%%%%%%%%%%%%%%
%%%% production firm
%%%%%%%%%%%%%%

% (10) labor demand
w = mc*(1-alpha)*A*(K(-1)/H)^(alpha);

% (11) Capital demand
R = mc*alpha*A*(K(-1)/H)^(alpha-1);

% (12) reset price
Pir = (epsi/(epsi-1))*x1/x2;

% (13) x1
x1 = Lam*mc*Y + thetap*beta*Pi(+1)^(epsi)*x1(+1);

% (14) x2
x2 = Lam*Y + thetap*beta*Pi(+1)^(epsi-1)*x2(+1);

%%%%%%%%%%%%%%
%%%% capital producer
%%%%%%%%%%%%%%

% (15) new capital
Ih = mu*(1 - (psik/2)*(I/I(-1) - 1)^2)*I;

% (16) FOC investment
pk*mu*(1 - (psik/2)*(I/I(-1) - 1)^2 - psik*(I/I(-1) - 1)*(I/I(-1))) = 1 -
beta*(Lam(+1)/Lam)*pk(+1)*mu(+1)*psik*(I(+1)/I - 1)*(I(+1)/I)^2;

%%%%%%%%%%%%%%
%%%% financial intermediary
%%%%%%%%%%%%%%

% (17) Balance sheet condition
Q*f + Q*b = L*N;

% (18) Net worth FOC
Lam*(1 + (psin/2)*(N/Ns - 1)^2 + psin*(N/Ns - 1)*N) =
beta*zeta*Lam(+1)*Pi(+1)^(-1)*((RL(+1) - Rd)*L + Rd);

% (19) Leverage condition
```

```
L = Lam(+1)*Pi(+1)^(-1)/(Lam(+1)*Pi(+1)^(-1) + (Phi -
1)*Lam(+1)*Pi(+1)^(-1)*(RL(+1)/Rd));

%%%%%%%%%%%%%%
%%%% monetary policy
%%%%%%%%%%%%%%

% (20) Taylor rule
log(Rd) = (1-rhoR)*log(Rds) + rhoR*log(Rd(-1)) + (1-rhoR)*(taupi*log(Pi) + tauy*(log(Y)
- log(Y(-1)))) + sR*eR;

%%%%%%%%%%%%%%
%%%% aggregate conditions
%%%%%%%%%%%%%%

% (21) Resource
Y = C + I + (psin/2)*(N/Ns - 1)^2*N;

% (22) Price evolution
1 = (1-thetap)*Pir^(1-epsi) + thetap*Pi^(epsi-1);

% (23) Wage evolution
w^(1-epsw) = (1-thetaw)*wr^(1-epsw) + thetaw*Pi^(epsw-1)*w(-1)^(1-epsw);

% (24) Production
dp*Y = A*K(-1)^(alpha)*H^(1-alpha);

% (25) Price dispersion
dp = (1-thetap)*Pir^(-epsi) + thetap*Pi^(epsi)*dp(-1);

% (26) Real government bonds
log(b) = (1-rhob1 - rhob2)*log(bs) + rhob1*log(b(-1)) + rhob2*log(b(-2)) - sb*eb;

% (27) Productivity
log(A) = rhoA*log(A(-1)) + sA*eA;

% (28) MEI
log(mu) = rhom*log(mu(-1)) + sm*em;

% (29) Credit shock
log(Phi) = (1-rhop)*log(PHIs) + rhop*log(Phi(-1)) + sp*ep;

%%%%%%%%%%%%%%
%%%% bond returns and yields
%%%%%%%%%%%%%%

% (30) long bond return
RL = (1+kappa*Q)/Q(-1);

% (31) YTM on long bond
Ry = Q^(-1) + kappa;

% (32) Hypothetical expectations hypothesis bond
QEH = (1+kappa*QEH(+1))/Rd;

% (33) YTM on EH bond
RyEH = QEH^(-1) + kappa;

% (34) Term premium
TP = Ry/RyEH;

%%%%% logged variables
logY = log(Y);

logRd = log(Rd);

logI = log(I);

logC = log(C);
```

```
logPi = log(Pi);
logTP = log(TP);
logRy = log(Ry);
logQ = log(Q);

end;

initval;
A = 1;
mu = 1;
Phi = PHIs;
Y = Ys;
C = Cs;
I = Is;
Ih = Is;
w = ws;
wr = ws;
Pi = 1;
Pir = 1;
b = bs;
f = fs;
K = Ks;
R = Rs;
RL = RLs;
Ry = Rys;
QEH = QEHs;
RyEH = RyEHs;
TP = TPs;
Lam = Lams;
L = Ls;
H = 1;
M = Ms;
pk = 1;
Q = Qs;
dp = 1;
N = Ns;
mc = mcs;
x1 = Lams*mcs*Ys/(1-thetap*beta);
x2 = Lams*Ys/(1-thetap*beta);
f1 = B/(1-thetaw*beta);
f2 = Lams/(1-thetaw*beta);
Rd = Rds;
logY = log(Ys);
logC = log(Cs);
logI = log(Is);
logPi = log(1);
logRd = log(Rds);
logRy = log(Rys);
logQ = log(Qs);
% var C H I Ih K Y mc N Rd R Q pk w wr Pi Pir f b f1 f2 x1 x2 L dp A mu Phi RL Ry QEH
RyEH TP Lam M;


end;

shocks;
var eA = 1;
var eR = 1;
var ep = 1;
var em = 1;
var eb = 1;
end;

stoch_simul(order=1,irf=20,nograph,ar=1);
```